DATA ACQUISITION FROM BLAST OVERPRESSURE TRIALS

D.R. KIRK

MRL-TN-628

MARCH 1993

MATERIALS RESEARCH LABORATORY

DSTO

Data Acquisition from Blast Overpressure Trials

D.R. Kirk

MRL Technical Note
MRL-TN-628

Abstract

A Macintish computer has been used to acquire data from blast overpressure trials on various weapons. The computer is connected to a multiple channel FM data recorder via a MacSCSI488 bus controller, allowing the computer to control the recorder and to acquire data from it through an analog to digital converter. Detailed instructions are given for connecting the hardware and operating the software involved.

DSTO MATERIALS RESEARCH LABORATORY

Published by

Materials Research Laboratory
Cordite Avenue, Maribyrnong
Victoria, 3032 Australia

Telephone: (03) 246 8111
Fax: (03) 246 8999

AR No. 008-234

APPROVED FOR PUBLIC RELEASE

Contents

Data Acquisition from Blast Overpressure Trials

1. Introduction

The firing of guns and other explosive weapons produces a rapid change in pressure in the atmosphere. These pressure changes can damage the hearing of any person close to the weapon, particularly the operator. To assess levels of overpressure, trials are performed at various sites, depending on the nature of the weapon being tested. The pressures are measured at a number of locations using pressure transducers to convert pressure to voltage which is recorded on a high speed FM data tape recorder. A typical recorder can store 14 channels of data at a speed of 76 cm/s. A playback speed of 1.2 cm/s gives a time magnification of about 63. Thus 1 ms of real time expands to 63 ms of tape time, allowing for much greater detail to be examined.

Records from the FM data recorder can be transferred to an oscilloscope and plotted with a pen recorder. Alternatively, the tape can be played through an analog to digital (A-to-D) converter and data transferred to a computer for a more detailed analysis. An arrangement that has been used for some time in Explosives Ordnance Division, Salisbury consists of transferring the data to a Macintosh computer via a MacSCSI488 Iotech Bus Controller. This is simply a device that converts IEEE488 interface signals to Small Computer Systems Interface (SCSI) signals and vice versa, thus allowing the Macintosh to communicate with IEEE devices, such as a data recorder and an A-to-D converter. The advantage of the Macintosh is the variety of software that is available to manipulate data and the convenient user interface. The software suite currently used to produce final pressure-time data includes a locally written program using Microsoft Quickbasic, to effect the transfer of data and also to manipulate the data acquired; McSink, a Macintosh desk accessory that allows hidden characters (such as carriage controls) to be edited; and Igor, a data manipulation and graph plotting program from WaveMetrics.

2. *Software*

A program written in Microsoft QuickBasic is used to transfer the data from the FM data recorder to the Macintosh via a 12-bit analog to digital converter. The program takes into account the voltage range used to record the data, the pressure to voltage conversion factor and the record and playback speeds. It prompts the user for information such as the file name to store a channel of data. It controls the operation of the recorder and the A-to-D converter, such as start and stop. All the operator has to do is to position the tape at an appropriate starting point and activate the program.

The data acquired from the FM data recorder are in a raw form and consist of columns of figures for time and pressure and a third column if a trigger channel is recorded. The zero point for the time base is arbitrary and if a time of arrival of a certain data value, for example maximum pressure, from a given point is required the data must be manipulated to reflect this. The WaveMetrics program "Igor" is ideal for this purpose. Data can be displayed and edited in tabular and graphic form. Factors can be applied to the data to adjust the time base to the correct zero point and a preliminary plot can be displayed on the screen. A Macintosh Desk Accessory McSink is used at a later time to further transform the data.

Some software is also required to drive the MacSCSI488 Interface. This is supplied by Iotech and must be installed according to the product manual.

3. *Hardware Connections*

Several pieces of hardware are needed to process the data. A Kyowa 14 channel RTP-650A data recorder is commonly used but any recorder can be substituted provided it has a minimum of 40kHz bandwith, FM mode, multiple speeds and IEEE (GPIB) Interface. An analog-to-digital converter such as the Kyowa ADC-116C is used to convert ± 5 volts signed input to a digital range of - 2048 to + 2048, an 11 bit range, with the twelfth bit indicating the sign. An Iotech MacSCSI488 bus controller converts IEEE signals to SCSI signals and vice versa to allow a Macintosh computer, typically an SE/30, to communicate with and control the recorder and the A-to-D converter. Subsequent discussion in this report applies to the hardware described but adaptations can be made for other hardware. Two IEEE488 cables are needed to connect the MacSCSI488 to the A-to-D converter and the recorder and a SCSI cable is used to connect the Macintosh to the MacSCSI488 bus controller. The data are transferred through one or two coaxial cables. To avoid the continual connecting and disconnecting of cables the pressure data are usually acquired by connecting the MONI output on the recorder to channel 1 of the A-to-D converter. This allows the operator to select the channel to be monitored by using the button on the recorder labelled "CHANNEL SELECT". If a trigger channel was recorded this channel is connected to the channel 2 input of the A-to-D converter. A diagram of the cable connections required is shown in Figure 1. Sometimes the trigger peak is a very narrow pulse and does not always register on the A-to-D converter, especially if the sampling rate is relatively slow (200 Hz for example). A small locally built electronic circuit is used to detect this peak and prolong it sufficiently to be recognised by the A

to D converter. This device is connected between the recorder and the A to D converter in the trigger channel coaxial cable.



Figure 1: Cable connections for transfer of data from VTR to Macintosh.

4. *Starting Up*

Before acquiring any data it has been found useful to record a copy of the original tape and use that in the recorder, rather than the original. A faulty recorder may damage the tape and data may be lost. A copy can be made by simply connecting the channel outputs from one data recorder to the channel inputs of another data recorder and playing the original tape on one recorder while recording on the other.

When starting up the various devices it is important to switch on the data recorder and the A-to-D converter first, then the MacSCSI488 bus controller, then the computer. The mouse button should be held down when switching on the computer and held until a list of INITs appears. One is called MacSCSI488 which must be selected as active. There must be no external drives on the SCSI interface, and any associated INITs should be switched off. It is also a good idea to disable any screen savers, which can cause loss of data on screen when the computer is acquiring data. Other INITs which might be disabled include TOPS and TOPS Spool but virus protection software such as Gatekeeper and Disinfectant should not be disabled. Finally, the computer should be operated in Finder and not Multifinder mode.

Once all the hardware requirements are complete, the tape cassette is inserted into the recorder and the required data found manually. The tape speed should be set at 1.2 cm/s and the tape positioned about 6 seconds before the first peak required. A simple procedure such as counting seconds is usually sufficiently accurate for this purpose. The compiled program is then activated to commence data acquisition.

5. *The Data Acquisition Program*

A listing of the QuickBasic program used to acquire data from one channel is given in Appendix I. Lines 18 to 24 handle the initialisation of the IEEE interface. The next section down to line 40 initialises program variables before displaying Window2. Lines 44 to 59 display the input window, buttons and default values set by the program. This window is displayed until the pointer is clicked on the CONTINUE button. The next section down to Window3 changes the program variables to reflect what the user has entered on the input window. Window3 displays a progress report on how many records are left to transfer. Statements beginning 'IeeeWrite' send various commands to the IEEE interface, such as telling the recorder to start. For a detailed description of what each command does the Kyowa manuals should be consulted. Line 117 converts the voltage from the recorder to kiloPascals (kPa) and computes the time elapsed since the first record was acquired. The factor 6/5 is used when a backup tape is being scanned. Backup tapes are recorded with an output range of 5 Volts and an input range of 6 Volts, hence the 6/5 conversion. If the original tape is used (although this is not recommended) this factor should be removed and the program recompiled. Lines 129 onwards display a termination window allowing the operator to continue with another record or to quit. The time taken for the previous transfer is also displayed. The subroutines from line 152 onwards are supplied by Iotech and are used to access the IEEE interface.

Appendix II is a listing of a similar program to acquire data when a trigger channel has been recorded as well.

6. *Operating the Acquisition Program*

The Microsoft QuickBasic environment allows compilation of programs to produce an executable Macintosh module. It is this compiled version, executed by double clicking on its Icon, which displays on the screen a window similar to Figure 2. The contents of the five rectangular boxes can be edited using the click and drag function of the Macintosh. The OUTPUT FILE NAME can be any legal Macintosh name. The Press (kPa/volt) box is the calibration value of the pressure transducer used. VTR Range is the voltage input range used when the data were originally acquired, usually somewhere between 0.4 and 6 volts. The No.of Points can be anything from about 2000 to 12 000 but 6000 is a good working compromise between taking a long time and risking missing the peak. A limited number of options is given for the Time interval, this is the interval that the A-to-D Converter will use to acquire data from the recorder. The user can enter 0.5, 1, 2 or 5 ms. These give sample rates of 2000, 1000, 500 and 200 Hz respectively. Any figure greater than 5 is set to 5, any figure less than 0.5 is set to 0.5. Other numbers will be converted to one of the four possible intervals. When converted from the recording speed of 76 cm/s to the playback speed of 1.2 cm/s these intervals convert to real time intervals from 7.9 to 79 µs. It is possible to use other values with a small program change but these have been found to be of limited use.



Figure 2: Data input window for data acquisition program.

Clicking on the CONTINUE button on the screen will resume program execution. The recorder should start and after a delay of about 5 seconds the A-D light on the A-to-D converter will come on. The Macintosh screen will change to a window similar to Figure 3. The operator should observe the monitored value in the top right hand corner of the recorder screen to verify that a peak has arrived. If this fails to happen, the operator should wait until the recorder stops and returns to LOCAL mode and should then stop the program by dragging the File menu item on the Macintosh to QUIT. The Macintosh will return to the Finder. The tape should then be repositioned and the program restarted. If the operator is satisfied that the peak has been acquired he can leave the machines for a few minutes until the Macintosh displays a window similar to Figure 4. Clicking on the NEW RECORD button will return the program to the input window. Clicking on the QUIT button will terminate the program.

When a few records have been transferred (for instance, all the channels for one shot) the program can be stopped and the acquired files can be copied to another computer, either via disk or local network. These files can be manipulated while more data is being transferred from the recorder to the computer. It takes 3 or 4 minutes to transfer 6000 points into one file and this time can be used to prepare the already acquired data for graphical display and printing.



Figure 3: Macintosh window while data is being transferred.



Figure 4: Termination window for data acquisition program.

7. Manipulating the Data

The data already acquired can be quickly scanned by using Igor. It is not within the scope of this Technical Note to describe in detail the use of Igor and only brief explanations are given. On opening Igor the Macintosh screen appears as shown in Figure 5. Pulling down the FILE menu to LOAD WAVES ... LOAD GENERAL TEXT will prompt the user for a filename. On opening the required file the user will be prompted to supply two or three Wave names. Data are stored in the order Times, Pressures then Trigger, if recorded. Pulling down the WAVES menu item to DISPLAY WAVES will prompt the user to select the X and Y variables to plot. The plot should be similar to Figure 6. Editing the data can be accomplished by pulling down the WAVES menu to EDIT WAVES and selecting all the waves displayed in the list. A portion of the data can be kept by highlighting the range in the table that is not wanted and using the CUT function (Command-X on the keyboard) to remove those values. In addition, if times of arrival are required, the arrival time of the trigger can be subtracted from the time values using a simple arithmetic expression. If the trigger time is 31.4 ms and the time data is labelled Times, typing in the Igor Command Window Times=Times-31.4 will alter all the time values by subtracting 31.4. This edited data should then be stored in a Macintosh file in text format. Dragging the FILE menu to SAVE WAVES ... SAVE IGOR TEXT will prompt the user to select the waves to be stored. Only the Time and Pressure should be stored as the trigger is no

longer required. The user will be prompted for a file name to store the data. The Igor program inserts text and unwanted TABS in the file. The text can simply be deleted and the TABS converted to spaces using the Macintosh Desk Accessory McSink. TABS are not handled correctly by Microsoft QuickBasic when reading input files. While editing using McSink, it is also necessary to insert the number of data values as the first line of the data. This is the same as No.of Points in Section 6 and is required by the next program Zerocorr, which is used to compute a mean zero value for the pressure data and operates by prompting the user for up to eight input and output files. The input window is similar to Figure 7. All the rectangular boxes can be edited using Macintosh click and drag procedures. Where less than eight files are availabe the first input box after the last file should be set to blank. Only files before the blank will be processed. The data is then ready for final output, using Igor to add such things as titles, peak heights and other useful information and to send a copy to a printer. It has been common practice to store the raw data and the Igor graph on floppy disk at this time and also on a backup medium. A 10 Mbyte Jasmine drive has been used for this purpose, however, the Macintosh must be shut down to connect the Jasmine drive or another computer used.



Figure 5: Igor window when first opened.



Figure 6: Igor graph of pressure vs time.



Figure 7: Input window for zero correction program.

8. Conclusion

A simple data acquisition system has been developed from existing equipment with the exception of the MacSCSI488 interface which was acquired at a small, once only cost. The system is simple to use and, with a copy of this report to hand, personnel should be able to operate it in a relatively short time.

Appendix I - Listing of Program for One Channel Acquisition

```
00001  '-------------------------------------------------------------------------
00002  '                    DATA TRANSFER FROM VTR TO MACINTOSH - ONE CHANNEL
00003  '-------------------------------------------------------------------------
00004  '
00005  '   TRANSFER PRESSURE DATA FROM KYOWA VTR TO MACINTOSH
00006  '   STORES DATA POINTS IN A FILE.  THIS VERSION CAN BE COMPILED
00007  '         DATE:           8/5/92
00008  '         FILENAME:    General.1chan
00009  '         AUTHOR:      D.R.KIRK
00010  '         LANGUAGE:  MICROSOFT QUICKBASIC
00011  '         INSTRUMENTS:   KYOWA A-D CONVERTER ADC-116C (IEEE ADDRESS 10)
00012  '                                KYOWA VTR RTP650A (IEEE ADDRESS 15)
00013  '                                MACINTOSH COMPUTER, ANY WITH SCSI PORT
00014  '         INTERFACE:        IOTECH MAC SCSI488 INTERFACE
00015  '-------------------------------------------------------------------------
00016  '        Ieee Initialization
00017  OPTION BASE 0
00018  LIBRARY "Device.LIB"                'Library of Basic routines
00019  IEEE%=0                                        'Interface buffers
00020  BufLen%=64
00021  Buf$=SPACE$(BufLen%)
00022  DIM ControlParams%(10)
00023  '-------------------------------------------------------------------------
00024  IeeeInit
00025  'Clear and localise AtoD & VTR
00026  IeeeWrite "CLEAR 10,15":         IeeeWrite "LOCAL 10,15"
00027
00028 '-------------------------------------------------------------------------
00029                              'Restart here if more data to analyse
00030  Newstart:
00031                               'Set initial values
00032  Npts%=6000                       'Number of data values
00033  Outrange=5!                         'VTR output range (Volts)
00034  Playback=1.2                       'VTR speed cm/s
00035  Record=76!                          'VTR speed cm/s
00036  Adinterval=.5                  'AD Converter interval (ms)
00037  Range=5!                             'AD Voltage range
00038  channel$="File01.01"           'File name for data storage
00039  chncal=7.3                                'Pressure kPa/Volt
00040  Inrange=1.4                              'VTR input range
00041 '-------------------------------------------------------------------------
00042  Window2:           ' display initial values in edit fields
00043
00044  WINDOW 2 ,"PROGRAM PARAMETERS",(0,40)-(525,336)
00045  BUTTON 5,1,"Alter or Verify these parameters",(140,20)-(375,35),1
00046  BUTTON 6,1,"Click on 'CONTINUE' when ready",(140,40)-(375,55),1
00047  EDIT FIELD 11,"Press (kPa/Volt)",(105,120)-(260,135),3
00048  EDIT FIELD 12,STR$(chncal),(270,120)-(400,135)
00049  EDIT FIELD 13,"VTR Range",(105,140)-(260,155),3
00050  EDIT FIELD 14,STR$(Inrange),(270,140)-(400,155)
00051  EDIT FIELD 15,"No.of Points (<= 12000)",(105,160)-(260,175),3
00052  EDIT FIELD 16,STR$(Npts%),(270,160)-(400,175)
00053  EDIT FIELD 17,"Time interval (ms)",(105,180)-(260,195),3
00054  EDIT FIELD 18,STR$(Adinterval),(270,180)-(400,195)
00055  EDIT FIELD 9,"OUTPUT FILE NAME",(105,100)-(260,115),3
00056  EDIT FIELD 10,channel$,(270,100)-(400,115)
00057  BUTTON 1,1,"CONTINUE",(300,245)-(370,260),1
00058  ACTIVITY=DIALOG(0)
00059  WHILE ACTIVITY<>1:ACTIVITY=DIALOG(0):WEND
00060  '-------------------------------------------------------------------------
00061  '    Extract altered values from edit fields
00062
00063  Irnge%=INT(Range)/4
00064  channel$=EDIT$(10)
00065  chncal=VAL(EDIT$(12))
00066  Inrange=VAL(EDIT$(14))
00067  Npts%=VAL(EDIT$(16))
00068  Adinterval=VAL(EDIT$(18))
00069  Runtime=Npts%*Adinterval*.001+2
00070  Timeint=Adinterval*Playback*.001/Record
```

```
00071  IF Adinterval<1 THEN
00072      Clk$="CLK6"
00073      GOTO Window3
00074  END IF
00075  IF Adinterval<1.8 THEN
00076      Clk$="CLK7"
00077      GOTO Window3
00078  END IF
00079  IF Adinterval<4 THEN
00080      Clk$="CLK8"
00081      GOTO Window3
00082  END IF
00083  Clk$="CLK9"
00084  LOCATE 15,10
00085  Window3:  '-----------------------------------------------------------------
00086  '                    Transfer data from VTR to A to D Converter
00087
00088  TZT=TIMER
00089  WINDOW CLOSE 2
00090  WINDOW 3,"***   TRANSFERRING DATA    ***",(0,40)-(525,336)
00091  BUTTON 1,1,"DATA FROM VTR TO AD CONV",(170,25)-(355,40),1
00092  IeeeWrite "OUTPUT 15;PF,EX          '15 is GPIB address of VTR, start VTR
00093  '       Reset AtoD, storage mode, rate
00094  IeeeWrite "OUTPUT 10;RST,MOD1,"+Clk$     '10 is address of AtoD converter
00095  '       Set for channel 1, range 1 or 5 volts
00096  IeeeWrite "OUTPUT 10;CHN001001,RNG"+MID$(STR$(Irnge%),2,1)
00097  ZTIME=TIMER
00098  WHILE TIMER-ZTIME<5:   WEND         'Delay A to D conversion to allow VTR
00099  IeeeWrite "OUTPUT 10;TRM0,STA"         'to come up to speed
00100  ZTIME=TIMER
00101  WHILE TIMER-ZTIME<Runtime:            WEND
00102  IeeeWrite "OUTPUT 10;STP"                    'Stop AtoD conversion
00103  IeeeWrite "OUTPUT 15;SP,EX"                  'Stop VTR
00104  IeeeWrite "CLEAR15":         IeeeWrite "LOCAL15"
00105  OPEN channel$ FOR OUTPUT AS #1
00106  BUTTON 1,1,"DATA FROM AD CONV TO FILE",(170,25)-(355,40),1
00107  '      Set AtoD to start reading at chn1, all channels, number of points
00108  IeeeWrite "OUTPUT 10;SPN1,RCH0,RDD"+MID$(STR$(Npts%),2)
00109  LOCATE 6,23
00110  PRINT " Records remaining"
00111  limit%=100
00112  k%=0
00113      WHILE k%<Npts%
00114          FOR i%=1 TO limit%
00115          k%=k%+1
00116          A$="":  IeeeWrite "ENTER 10#6":   IeeeRead A$:        A1%=VAL(A$)
00117          Pressure=A1%*chncal*Range*Inrange/(2000*Outrange)
00118          Time1=(k%-1)*Timeint
00119          PRINT #1,Time1,Pressure*6/5      '6/5 only used for backup tape
00120          NEXT i%
00121      LOCATE 6,40
00122      PRINT Npts%-k%;"          "
00123      WEND
00124  IeeeWrite "CLEAR10":         IeeeWrite "LOCAL10"
00125  CLOSE #1
00126  LOCATE 6,40
00127  PRINT "            "
00128  '-----------------------------------------------------------------------
00129  Finis:
00130  TOTIME=TIMER-TZT
00131  BUTTON 3,1,"WARNING!, REWIND VTR TO START POINT BEFORE
                          PROCEEDING",(50,230)-(450,245),1
00132  BUTTON 1,1,"QUIT",(80,260)-(120,275),1
00133  BUTTON 2,1,"NEW RECORD",(150,260)-(250,275),1
00134  LOCATE 12,20
00135  PRINT "TIME TAKEN = ";TOTIME;" SECONDS"
00136  ACTIVITY=DIALOG(0)
00137      WHILE ACTIVITY<>1
00138      ACTIVITY=DIALOG(0)
00139      WEND
00140  ON DIALOG(1) GOTO Termin,Newst,Finis
00141  Newst:
00142  BUTTON CLOSE 1
00143  BUTTON CLOSE 2
```

```
00144  BUTTON CLOSE 3
00145  WINDOW CLOSE 3
00146  GOTO Window2
00147  Termin:
00148  WINDOW CLOSE 3
00149  Quits:
00150  END
00151  '-------------------------------------------------------------------------
Note:  The following subroutines appear at the end of both programs, although
  they are omiited from the second listing.
00152  '     Ieee SubRoutines required by Iotech MacSCSI488 bus controller
00153  SUB IeeeInit STATIC
00154    SHARED IEEE%
00155    written%=0
00156    OpenDriver ".IEEE",IEEE%
00157    IeeeRestart
00158    IeeeWrite "RESET"
00159    IeeeWrite "EOL IN NONE"
00160  END SUB
00161
00162  SUB IeeeFinish STATIC
00163    SHARED IEEE%
00164    CloseDriver IEEE%
00165  END SUB
00166
00167  SUB IeeeRestart STATIC
00168    SHARED IEEE%,ControlParams%()
00169    Control IEEE%,256,ControlParams%(0)
00170  END SUB
00171
00172  SUB IeeeState(State%) STATIC
00173    SHARED IEEE%,ControlParams%()
00174    Status IEEE%,0,ControlParams%(0)
00175    State%=ControlParams%(0)
00176  END SUB
00177
00178  SUB IeeeActive(State%) STATIC
00179    SHARED IEEE%,ControlParams%()
00180    Status IEEE%,0,ControlParams%(0)
00181    State%=ControlParams%(1)
00182  END SUB
00183
00184  SUB IeeeWriteB(text$) STATIC
00185    SHARED IEEE%
00186    written%=0
00187    FSWrite IEEE%,text$,written%
00188  END SUB
00189
00190  SUB IeeeWrite(text$) STATIC
00191    IeeeWriteB(text$+CHR$(13))
00192  END SUB
00194  SUB IeeeRead(result$) STATIC
00195    SHARED IEEE%,Buf$
00196    ReadIn%=0
00197    FSRead IEEE%,Buf$,ReadIn%
00198    result$=LEFT$(Buf$,ReadIn%)
00199    IeeeState State%
00200    WHILE State%<>0
00201      FSRead IEEE%,Buf$,ReadIn%
00202      result$=result$+LEFT$(Buf$,ReadIn%)
00203      IeeeState State%
00204    WEND
00205  END SUB
```

Appendix II - Listing of Program for Two Channel Acquisition

```
00001  '-------------------------------------------------------------------------
00002  '               DATA TRANSFER FROM VTR TO MACINTOSH - TWO CHANNEL
00003  '-------------------------------------------------------------------------
00004  '
00005  '  TRANSFER KYOWA VTR DATA AND TRIGGER DATA TO A MACINTOSH FILE
00006  '    DATE:            8/5/92/91
00007  '         FILENAME:    General.trig
00008  '         AUTHOR:      D.R.KIRK
00009  '         LANGUAGE:  MICROSOFT QUICKBASIC
00010  '         INSTRUMENTS:   KYOWA A-D CONVERTER ADC-116C
00011  '                                KYOWA VTR RTP650A
00012  '                                MACINTOSH SE OR SE/30
00013  '        INTERFACE:        MACSCSI 488   -   IOTECH
00014  '-------------------------------------------------------------------------
00015  '        Ieee Initialization
00016  OPTION BASE 0
00017  LIBRARY "Device.LIB"
00018  IEEE%=0
00019  BufLen%=64
00020  Buf$=SPACE$(BufLen%)
00021  DIM ControlParams%(10)
00022  '-------------------------------------------------------------------------
00023  IeeeInit
00024  IeeeWrite "CLEAR 10,15":         IeeeWrite "LOCAL 10,15"
00025                              'Clear                and localise AtoD & VTR
00026
00027  '-------------------------------------------------------------------------
00028  channel$="File01.01"                    'File name for data storage
00029  Npts%=6000
00030  Inrange=6                                'VTR input range
00031  Outrange=5!                             'VTR output range
00032  Playback=1.2                           'VTR cm/s
00033  Record=76!                              'VTR cm/s
00034  Adinterval=1!                    'AD Converter (ms)
00035  Range=5!                                  'AD Voltage range
00036  Irnge%=INT(Range)/4
00037  chncal=7!
00038  Newstart:                       'Restart here if more data to analyse
00039  Window2:
00040  WINDOW 2 ,"PROGRAM PARAMETERS",(0,40)-(525,336)
00041  BUTTON 5,1,"Alter or Verify these parameters",(140,20)-(375,35),1
00042  BUTTON 6,1,"Click on 'CONTINUE' when ready",(140,40)-(375,55),1
00043  EDIT FIELD 11,"Pressure (kPa/Volt)",(105,120)-(250,135),3
00044  EDIT FIELD 12,STR$(chncal),(260,120)-(400,135)
00045  EDIT FIELD 13,"VTR Range",(105,140)-(250,155),3
00046  EDIT FIELD 14,STR$(Inrange),(260,140)-(400,155)
00047  EDIT FIELD 15,"Time Interval (ms)",(105,160)-(250,175),3
00048  EDIT FIELD 16,STR$(Adinterval),(260,160)-(400,175)
00049  EDIT FIELD 17,"No.of Points (<8000)",(105,180)-(250,195),3
00050  EDIT FIELD 18,STR$(Npts%),(260,180)-(400,195)
00051  EDIT FIELD 9,"OUTPUT FILE NAME",(105,100)-(250,115),3
00052  EDIT FIELD 10,channel$,(260,100)-(400,115)
00053  BUTTON 1,1,"CONTINUE",(300,230)-(370,245),1
00054  ACTIVITY=DIALOG(0)
00055  WHILE ACTIVITY<>1:ACTIVITY=DIALOG(0):WEND
00056  channel$=EDIT$(10)
00057  chncal=VAL(EDIT$(12))
00058  Inrange=VAL(EDIT$(14))
00059  Adinterval=VAL(EDIT$(16))
00060  Npts%=VAL(EDIT$(18))
00061  Runtime=Npts%*Adinterval*.001+2
00062  Timeint=Adinterval*Playback/Record
00063  IF Adinterval<1 THEN
00064      Clk$="CLK6"                 ' Ad interval = 0.5 ms  Real time 7.9 us
00065      GOTO Window3
00066  END IF
00067  IF Adinterval<1.8 THEN
00068      Clk$="CLK7"                 ' Ad interval = 1.0 ms Real time 15.8 us
00069      GOTO Window3
```

```
00070  END IF
00071  IF Adinterval<4 THEN
00072      Clk$="CLK8"                ' Ad interval = 2.0 ms Real time 31.7 us
00073      GOTO Window3
00074  END IF
00075  Clk$= "CLK9"                   ' Ad interval = 5.0 ms Real time 78.9 us
00076  LOCATE 15,10
00077  Window3: '-----------------------------------------------------------------
00078  TZT=TIMER
00079  WINDOW CLOSE 2
00080  WINDOW 3,"***   TRANSFERRING DATA    ***",(0,40)-(525,336)
00081  BUTTON 1,1,"DATA FROM VTR TO AD CONV",(170,25)-(355,40),1
00082  IeeeWrite "OUTPUT 15;PF,EX"
00083  IeeeWrite "OUTPUT 10;RST,MOD1,"+Clk$                ' Storage Mode,
Interval
00084  IeeeWrite "OUTPUT 10;CHN001002,RNG"+MID$(STR$(Irnge%),2,1)
00085  ztime=TIMER
00086  WHILE TIMER-ztime<5: WEND                    'delay A to D start by 5
second
00087  IeeeWrite "OUTPUT 10;TRM0,STA"                       'Start A to D
conversion
00088  ztime=TIMER
00089  WHILE TIMER-ztime<Runtime:            WEND
00090  IeeeWrite "OUTPUT 15;SP,EX"                          'Stop VTR
00091  IeeeWrite "OUTPUT 10;STP"                            'Stop A to D
converter
00092  IeeeWrite "CLEAR15":    IeeeWrite "LOCAL15"    'Clear and localise VTR
00093  BUTTON 1,1,"DATA FROM AD CONV TO FILE",(170,25)-(355,40),1
00094  IeeeWrite "OUTPUT 10;SPN1,RCH0,RDD"+MID$(STR$(Npts%),2)
                            'Set A to D for data output
00095  OPEN channel$ FOR OUTPUT AS #1
00096  LOCATE 4,23
00097  PRINT " Records remaining"
00098  limit%=100
00099  k%=0
00100      WHILE k%<Npts%
00101          FOR i%=1 TO limit%
00102          k%=k%+1
00103          A$="":    IeeeWrite "ENTER 10#6":     IeeeRead A$:
A1%=VAL(A$)
00104          B$="":    IeeeWrite "ENTER 10#6":     IeeeRead B$:
B1%=VAL(B$)
00105          Pressure=A1%*chncal*Range*Inrange/(2000*Outrange)
00106          PRINT #1,k%*Timeint*.001;CHR$(9);Pressure*6/5;CHR$(9);B1%
00107                                                  '6/5 only for backup
tape
00108          NEXT i%
00109      LOCATE 4,40
00110      PRINT Npts%-k%;"          "
00111      WEND
00112  LOCATE 4,40
00113  PRINT "             "
00114  IeeeWrite "CLEAR10":        IeeeWrite "LOCAL10"     'Clear and localise
AtoD
00115  CLOSE #1
00116  '-----------------------------------------------------------------------
00117  Finis:
00118  TOTIME=TIMER-TZT
00119  BUTTON 6,1,"WARNING!, REWIND VTR TO START POINT BEFORE
                          PROCEEDING",(50,230)-(450,245),1
00120  BUTTON 1,1,"QUIT",(80,260)-(120,275),1
00121  BUTTON 2,1,"NEW RECORD",(150,260)-(250,275),1
00122  LOCATE 12,20
00123  PRINT "TIME TAKEN = ";TOTIME;" SECONDS
00124  ACTIVITY=DIALOG(0):  WHILE ACTIVITY<>1: ACTIVITY=DIALOG(0):  WEND
00125  ON DIALOG(1) GOTO Termin,Newst
00126  Newst:
00127  BUTTON CLOSE 1
00128  BUTTON CLOSE 2
00129  BUTTON CLOSE 6
00130  WINDOW CLOSE 3
00131  GOTO Newstart
00132  Termin:
00133  WINDOW CLOSE 3
```

```
00134  Quits:
00135  END
00136 '-------------------------------------------------------------------
00137  '       Ieee functions as for one channel listing
```

SECURITY CLASSIFICATION OF THIS PAGE **UNCLASSIFIED**

REPORT NO.	AR NO.	REPORT SECURITY CLASSIFICATION
MRL-TN-628	AR-008-234	Unclassified

TITLE

Data acquisition from blast overpressure trials

AUTHOR(S)
D.R. Kirk

CORPORATE AUTHOR
DSTO Materials Research Laboratory
PO Box 50
Ascot Vale Victoria 3032

REPORT DATE	TASK NO.	SPONSOR
March, 1993	ARM 89/178	Army

FILE NO.	REFERENCES	PAGES
G6/4/8-4332	-	21

CLASSIFICATION/LIMITATION REVIEW DATE

CLASSIFICATION/RELEASE AUTHORITY
Chief, Explosives Ordnance Division

SECONDARY DISTRIBUTION

Approved for public release

ANNOUNCEMENT

Announcement of this report is unlimited

KEYWORDS

User manual | Data processing | Blast overpressure

ABSTRACT

A Macintish computer has been used to acquire data from blast overpressure trials on various weapons. The computer is connected to a multiple channel FM data recorder via a MacSCS1488 bus controller, allowing the computer to control the recorder and to acquire data from it through an analog to digital converter. Detailed instructions are given for connecting the hardware and operating the software involved.

Data Acquisition from Blast Overpressure Trials

D.R. Kirk

(MRL-TN-628)

DISTRIBUTION LIST

Director, MRL
Chief, Explosives Ordnance Division
Mr L. Sheppard
Mr D.R. Kirk
MRL Information Service

Chief Defence Scientist (for CDS, FASSP, ASSCM) (1 copy only)
Director, Surveillance Research Laboratory
Director (for Library), Aeronautical Research Laboratory
Director, Electronics Research Laboratory
Head, Information Centre, Defence Intelligence Organisation
OIC Technical Reports Centre, Defence Central Library
Officer in Charge, Document Exchange Centre (8 copies)
Army Scientific Adviser, Russell Offices
Air Force Scientific Adviser, Russell Offices
Navy Scientific Adviser, Russell Offices - data sheet only
Scientific Adviser, Defence Central
Director-General Force Development (Land)
Senior Librarian, Main Library DSTOS
Librarian, MRL Sydney
Librarian, H Block
UK/USA/CAN ABCA Armies Standardisation Rep. c/- DGAT (8 copies)
Librarian, Australian Defence Force Academy
Counsellor, Defence Science, Embassy of Australia - data sheet only
Counsellor, Defence Science, Australian High Commission - data sheet only
Scientific Adviser to DSTC, C/- Defence Adviser - data sheet only
Scientific Adviser to MRDC, C/- Defence Attache - data sheet only
Head of Staff, British Defence Research and Supply Staff (Australia)
NASA Senior Scientific Representative in Australia
INSPEC: Acquisitions Section Institution of Electrical Engineers
Head Librarian, Australian Nuclear Science and Technology Organisation
Senior Librarian, Hargrave Library, Monash University
Library - Exchange Desk, National Institute of Standards and Technology, US
Exchange Section, British Library Document Supply Centre
Periodicals Recording Section, Science Reference and Information Service, UK
Library, Chemical Abstracts Reference Service
Engineering Societies Library, US
Documents Librarian, The Center for Research Libraries, US

Dr R. Glass, EOD-S
Dr H.T. French, EOD-S
Mr J.K. Leach, EOD-S
Mr P.L. Fagan, EOD-S
Mr K.L. Krishnamoorthy, EOD-S
Mr G. Teague, EOD-S
Dr S. Stephenson, EOD-S
Mr J.S. Howe, SSD Melbourne